431Intro

last edited on March 11, 2013 09:18 PM by admin

File... Action Data... sage Typeset Print Worksheet Edit Text Undo Share Publish

```
f(x) = exp(2*x)/(1+exp(2*x))
d = derivative(f(x),x,7)
d
```

$$\frac{128\,e^{(2\,x)}}{e^{(2\,x)}+1} - \frac{16256\,e^{(4\,x)}}{(e^{(2\,x)}+1)^2} + \frac{247296\,e^{(6\,x)}}{(e^{(2\,x)}+1)^3} - \frac{1306368\,e^{(8\,x)}}{(e^{(2\,x)}+1)^4} + \frac{3225600\,e^{(10\,x)}}{(e^{(2\,x)}+1)^5} - \frac{4085760\,e^{(12\,x)}}{(e^{(2\,x)}+1)^6} + \frac{2580480\,e^{(14\,x)}}{(e^{(2\,x)}+1)^7} - \frac{645120\,e^{(16\,x)}}{(e^{(2\,x)}+1)^8}$$

```
f(x).derivative(x,7)
```

$$\frac{128\,e^{(2\,x)}}{e^{(2\,x)}+1} - \frac{16256\,e^{(4\,x)}}{(e^{(2\,x)}+1)^2} + \frac{247296\,e^{(6\,x)}}{(e^{(2\,x)}+1)^3} - \frac{1306368\,e^{(8\,x)}}{(e^{(2\,x)}+1)^4} + \frac{3225600\,e^{(10\,x)}}{(e^{(2\,x)}+1)^5} - \frac{4085760\,e^{(12\,x)}}{(e^{(2\,x)}+1)^6} + \frac{2580480\,e^{(14\,x)}}{(e^{(2\,x)}+1)^7} - \frac{645120\,e^{(16\,x)}}{(e^{(2\,x)}+1)^8}$$

```
show(d.factor()) # Bigger Typeface
```

$$-\frac{128\left(120\,e^{(2\,x)} - 1191\,e^{(4\,x)} + 2416\,e^{(6\,x)} - 1191\,e^{(8\,x)} + 120\,e^{(10\,x)} - e^{(12\,x)} - 1\right)e^{(2\,x)}}{\left(e^{(2\,x)} + 1\right)^8}$$

```
d(x=0)
```

$$-136$$

```
# MGF of normal
var('t mu sigma')
m = exp(mu*t + 1/2 * sigma^2 * t^2)
show(m)
```

$$e^{\left(\frac{1}{2}\,\sigma^2 t^2 + \mu t\right)}$$

```
EX9 = m.derivative(t,9)(t=0)
show(factor(EX9))
```

$$\left(\mu^8 + 36\,\mu^6\sigma^2 + 378\,\mu^4\sigma^4 + 1260\,\mu^2\sigma^6 + 945\,\sigma^8\right)\mu$$

```
# Get it directly
show(pi)
```

$$\pi$$

```
f = 1/(sigma*sqrt(2*pi)) * exp(-1/2*(x-mu)^2/sigma^2)
show(f)
```

$$\frac{\sqrt{2}e^{\left(-\frac{(\mu-x)^2}{2\,\sigma^2}\right)}}{2\,\sqrt{\pi}\sigma}$$

```
ex9 = integrate(x^9*f,x,-oo,oo)
ex9
```

```
Traceback (click to the left of this block for traceback)
...
Is  sigma  positive or negative?
```

```
assume(sigma>0)
ex9 = integrate(x^9*f,x,-oo,oo)
show(ex9)
```

$$\frac{\left(\sqrt{\pi}\sqrt{2}\mu^9\sigma + 36\,\sqrt{\pi}\sqrt{2}\mu^7\sigma^3 + 378\,\sqrt{\pi}\sqrt{2}\mu^5\sigma^5 + 1260\,\sqrt{\pi}\sqrt{2}\mu^3\sigma^7 + 945\,\sqrt{\pi}\sqrt{2}\mu\sigma^9\right)\sqrt{2}}{2\,\sqrt{\pi}\sigma}$$

```
show(factor(ex9))
```

$$\left(\mu^8 + 36\,\mu^6\sigma^2 + 378\,\mu^4\sigma^4 + 1260\,\mu^2\sigma^6 + 945\,\sigma^8\right)\mu$$

```
show(factor(EX9))
```

$$\left(\mu^8 + 36\,\mu^6\sigma^2 + 378\,\mu^4\sigma^4 + 1260\,\mu^2\sigma^6 + 945\,\sigma^8\right)\mu$$

```
# Matrix Calculations
# D = Lambda F + e
# V(F)=Phi, V(e) = Psi
# V(D) = Lambda Phi Lambda' + Psi
# Set up Matrices: SR means Symbolic Ring
# Indices start at Zero, not one
Phi = matrix(SR,3,3) # V(F), Symmetric
Phi[0,0] = var('phi11'); Phi[0,1] = var('phi12');  Phi[0,2] = var('phi13')
Phi[1,0] = var('phi12'); Phi[1,1] = var('phi22');  Phi[1,2] = var('phi23')
Phi[2,0] = var('phi13'); Phi[2,1] = var('phi23');  Phi[2,2] = var('phi33')
show(Phi)
```

$$\begin{pmatrix} \phi_{11} & \phi_{12} & \phi_{13} \\ \phi_{12} & \phi_{22} & \phi_{23} \\ \phi_{13} & \phi_{23} & \phi_{33} \end{pmatrix}$$

```
# Set up Lambda
Lambda = matrix(SR,6,3)
Lambda[0,0]= 1; Lambda[1,0] = var('lambda2')
Lambda[2,1]= 1; Lambda[3,1] = var('lambda4')
Lambda[4,2]= 1; Lambda[5,2] = var('lambda6')
show(Lambda)
```

$$\begin{pmatrix} 1 & 0 & 0 \\ \lambda_2 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & \lambda_4 & 0 \\ 0 & 0 & 1 \\ 0 & 0 & \lambda_6 \end{pmatrix}$$

```
# Set up Psi (use a Python loop)
Psi = matrix(SR,6,6)
for i in interval(1,6): Psi[i-1,i-1] = var('psi'+str(i))
show(Psi)
```

$$\begin{pmatrix} \psi_1 & 0 & 0 & 0 & 0 & 0 \\ 0 & \psi_2 & 0 & 0 & 0 & 0 \\ 0 & 0 & \psi_3 & 0 & 0 & 0 \\ 0 & 0 & 0 & \psi_4 & 0 & 0 \\ 0 & 0 & 0 & 0 & \psi_5 & 0 \\ 0 & 0 & 0 & 0 & 0 & \psi_6 \end{pmatrix}$$

```
# Calculate covariance matrix of observable data
Sigma = Lambda * Phi * Lambda.transpose() + Psi
show(Sigma)
```

evaluate

$$\begin{pmatrix} \phi_{11} + \psi_1 & \lambda_2\phi_{11} & \phi_{12} & \lambda_4\phi_{12} & \phi_{13} & \lambda_6\phi_{13} \\ \lambda_2\phi_{11} & \lambda_2^2\phi_{11} + \psi_2 & \lambda_2\phi_{12} & \lambda_2\lambda_4\phi_{12} & \lambda_2\phi_{13} & \lambda_2\lambda_6\phi_{13} \\ \phi_{12} & \lambda_2\phi_{12} & \phi_{22} + \psi_3 & \lambda_4\phi_{22} & \phi_{23} & \lambda_6\phi_{23} \\ \lambda_4\phi_{12} & \lambda_2\lambda_4\phi_{12} & \lambda_4\phi_{22} & \lambda_4^2\phi_{22} + \psi_4 & \lambda_4\phi_{23} & \lambda_4\lambda_6\phi_{23} \\ \phi_{13} & \lambda_2\phi_{13} & \phi_{23} & \lambda_4\phi_{23} & \phi_{33} + \psi_5 & \lambda_6\phi_{33} \\ \lambda_6\phi_{13} & \lambda_2\lambda_6\phi_{13} & \lambda_6\phi_{23} & \lambda_4\lambda_6\phi_{23} & \lambda_6\phi_{33} & \lambda_6^2\phi_{33} + \psi_6 \end{pmatrix}$$